CALCUL APPROCHÉ D’INTÉGRALES

Méthode des trapèzes

```
F1▾   F2▾     F3▾ F4▾ F5      F6▾
Tools Control I/O Var Find... Mode
:int_trap(f,a,b,n)
:Prgm
:ClrIO
:Local g,h,tmp,i,aire,c,d
:expr(string(f)&"→g(x)")
:approx(abs(b-a)/n)→h
:1→tmp
:0→aire
:
:Disp "MENU.............."
:Disp "1. AVEC graphique"
:Disp "2. SANS graphique"
:Disp ""
:Input "Votre choix ?",c
:ClrIO
:
:If c=2
:Goto suite
:
:© Création graphique
:g(zeros(d(g(x),x),x))→d
:a→xmin:b→xmax
:min(min(g(a),g(b)),min(d)
 )→ymin
:max(max(g(a),g(b)),max(d)
 )→ymax
:FnOff :PlotsOff
:
:expr(string(f)&"→y1(x)")
:DispG
:
:For i,1,n
:  Line a+(i-1)*h,0,a+(i-1
 )*h,g(a+(i-1)*h)
:  Line a+(i-1)*h,g(a+(i-1
 )*h),a+i*h,g(a+i*h)
:EndFor
:Line b,0,b,g(b)
:
:FnOn :PlotsOn
:Pause
:© Fin création graphique
:
:Lbl suite
:If a>b Then
:  b→tmp:a→b:tmp→a: -1→tmp
:EndIf
:
:For i,1,n
: approx(aire+h*(g(a+(i-1)
 *h)+g(a+i*h))/2)→aire
:EndFor
:tmp*aire→aire
:Disp "∫(f(x),x,"&string(a
 )&","&string(b)&") est","
 approximée par :",approx(
 aire),"(méthode des trapè
 zes)","","(calto : "&stri
 ng(approx(∫(g(x),x,a,b)))
 &")"
:EndPrgm
MATHS        RAD AUTO      FUNC
```

***f* définie sur l’intervalle fermé [*a*, *b*] subdivisé en *n* parties égales**

Attention, f n’est pas ici une fonction, juste sa valeur prise au point x. On place donc l’expression de f dans $g(x)$, de sorte que g soit une fonction utilisable.



Définition de la fenêtre graphique. Précisons que si D = $\{g(x) \mid x \in [a, b] \Rightarrow g'(x) = 0\} \cup \{g(a), g(b)\}$, alors $\min_{x\in[a,b]} g(x) = \min(D)$ (resp. $\max_{x\in[a,b]} g(x) = \max(D)$)

On place l’expression de f dans $y1(x)$, fonction de l’éditeur d’équations, que l’on trace avec `DispG`. Les lignes `FnOff :PlotsOff` désactivent les autres fonctions $y2, \dots, yn$ éventuellement présentes.

Tracé des trapèzes. Pour tout $i \in \{1, \dots, n\}$, on trace le trapèze de sommets $\big((a + (i-1)h), 0\big)$, $\big((a + (i-1)h), g(a + (i+1)h)\big)$, $\big(a + ih, g(a + ih)\big)$, $\big(a + ih, 0\big)$

Si $a > b$, on inverse les rôles de a et b, en multipliant simplement l’aire finale par $tmp = -1$ (autrement, $tmp = 1$)

Pour tout $i \in \{1, \dots, n\}$, l’aire du trapèze construit vaut $h/2\,\big(g(a+(i-1)h + g(a+ih)\big)$. On l’ajoute à la variable aire et on incrémente i.

On affiche le résultat de l’approximation par cette méthode, et ce que donne la calculatrice pour comparer.